```
In[•]:= << KnotTheory`
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
In[•]:= RotateToMinimal[l_] := Module[
          {bestl = l, rotatedl = RotateLeft[l]},
          While[rotatedl =!= l,
           bestl = First[Sort[{bestl, rotatedl}]];
           rotatedl = RotateLeft[rotatedl]
          ];
          bestl
         ];
```

```
In[•]:= pd = PD[Knot[11, Alternating, 53]]
```

```
KnotTheory: Loading precomputed data in DTCode4KnotsTo11`.
KnotTheory: The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of
            2005.
```

```
Out[•]= PD[X[4, 2, 5, 1], X[8, 3, 9, 4], X[14, 6, 15, 5],
          X[16, 7, 17, 8], X[2, 9, 3, 10], X[18, 11, 19, 12], X[20, 13, 21, 14],
          X[22, 16, 1, 15], X[10, 17, 11, 18], X[12, 19, 13, 20], X[6, 21, 7, 22]]
```

```
In[•]:= (Times @@
           (pd /. X[i_, j_, k_, l_] :> If[PositiveQ@X[i, j, k, l], c[i, j] c[l, k], c[i, l] c[j, k]]))
```

```
Out[•]= c[1, 5] c[2, 10] c[3, 9] c[4, 2] c[5, 15] c[6, 22] c[7, 17]
         c[8, 4] c[9, 3] c[10, 18] c[11, 19] c[12, 20] c[13, 21] c[14, 6] c[15, 1]
         c[16, 8] c[17, 11] c[18, 12] c[19, 13] c[20, 14] c[21, 7] c[22, 16]
```

```
In[•]:= cycles = (Times @@ (pd /. X[i_, j_, k_, l_] :>
                  If[PositiveQ@X[i, j, k, l], cyc[i, j] cyc[l, k], cyc[i, l] cyc[j, k]])) //.
          cyc[i_, m1___, j_] cyc[j_, m2___, k_] :> cyc[i, m1, j, m2, k] /.
         c_cyc :> RotateToMinimal@Most[c]
```

```
Out[•]= cyc[3, 9] cyc[1, 5, 15] cyc[7, 17, 11, 19, 13, 21] cyc[2, 10, 18, 12, 20, 14, 6, 22, 16, 8, 4]
```

```
In[•]:= cycles = SortBy[List @@ cycles, First]
```

```
Out[•]= {cyc[1, 5, 15], cyc[2, 10, 18, 12, 20, 14, 6, 22, 16, 8, 4],
         cyc[3, 9], cyc[7, 17, 11, 19, 13, 21]}
```

bl stands for “braid line”.

```
In[•]:= bl = cycles; Print[bl];
        While[Length[bl] > 1,
         m = bl[[-1, 1]];
         bl = Most[bl] /. m - 1 → Sequence @@ Prepend[Reverse@Last@bl, m - 1];
         Print[bl]
        ]
```

```
{cyc[1, 5, 15], cyc[2, 10, 18, 12, 20, 14, 6, 22, 16, 8, 4], cyc[3, 9], cyc[7, 17, 11, 19, 13, 21]}

{cyc[1, 5, 15], cyc[2, 10, 18, 12, 20, 14, 6, 21, 13, 19, 11, 17, 7, 22, 16, 8, 4], cyc[3, 9]}

{cyc[1, 5, 15], cyc[2, 9, 3, 10, 18, 12, 20, 14, 6, 21, 13, 19, 11, 17, 7, 22, 16, 8, 4]}

{cyc[1, 4, 8, 16, 22, 7, 17, 11, 19, 13, 21, 6, 14, 20, 12, 18, 10, 3, 9, 2, 5, 15]}
```

In[ ]:=
```
{1, 2, 3, 4} /. 3 → {5, 6}
```

Out[ ]=
```
{1, 2, {5, 6}, 4}
```

In[ ]:=
```
{1, 2, 3, 4} /. 3 → Sequence @@ {5, 6}
```

Out[ ]=
```
{1, 2, 5, 6, 4}
```